

#include <machine/asm.h>
#include "atomic_op_asm.h"

	.text
	.set	noreorder
#ifdef _KERNEL_OPT
#include "opt_cputype.h"
#ifndef MIPS3_LOONGSON2F
	.set	noat
	.set	nomacro
#endif
#else /* _KERNEL_OPT */
	.set	noat
	.set	nomacro
#endif /* _KERNEL_OPT */


LEAF(_atomic_and_32)
1:	INT_LL		t0, 0(a0)
	 nop
	and		t0, a1
	INT_SC		t0, 0(a0)
	beq		t0, zero, 1b
 	 nop
	j		ra
	 nop
END(_atomic_and_32)
ATOMIC_OP_ALIAS(atomic_and_32, _atomic_and_32)

LEAF(_atomic_and_32_nv)
1:	INT_LL		v0, 0(a0)
	 nop
	and		v0, a1
	move		t0, v0
	INT_SC		t0, 0(a0)
	beq		t0, zero, 1b
 	 nop
	j		ra
	 nop
END(_atomic_and_32_nv)
ATOMIC_OP_ALIAS(atomic_and_32_nv, _atomic_and_32_nv)

#if !defined(__mips_o32)
LEAF(_atomic_and_64)
1:	REG_LL		t0, 0(a0)
	 nop
	and		t0, a1
	REG_SC		t0, 0(a0)
	beq		t0, zero, 1b
 	 nop
	j		ra
	 nop
END(_atomic_and_64)
ATOMIC_OP_ALIAS(atomic_and_64, _atomic_and_64)

LEAF(_atomic_and_64_nv)
1:	REG_LL		v0, 0(a0)
	 nop
	and		v0, a1
	move		t0, v0
	REG_SC		t0, 0(a0)
	beq		t0, zero, 1b
 	 nop
	j		ra
	 nop
END(_atomic_and_64_nv)
ATOMIC_OP_ALIAS(atomic_and_64_nv, _atomic_and_64_nv)
#endif

#ifdef _LP64
STRONG_ALIAS(_atomic_and_ulong,		_atomic_and_64)
STRONG_ALIAS(_atomic_and_ulong_nv,	_atomic_and_64_nv)
#else
STRONG_ALIAS(_atomic_and_ulong,		_atomic_and_32)
STRONG_ALIAS(_atomic_and_ulong_nv,	_atomic_and_32_nv)
#endif
STRONG_ALIAS(_atomic_and_uint,		_atomic_and_32)
STRONG_ALIAS(_atomic_and_uint_nv,	_atomic_and_32_nv)

ATOMIC_OP_ALIAS(atomic_and_uint,	_atomic_and_uint)
ATOMIC_OP_ALIAS(atomic_and_uint_nv,	_atomic_and_uint_nv)
ATOMIC_OP_ALIAS(atomic_and_ulong,	_atomic_and_ulong)
ATOMIC_OP_ALIAS(atomic_and_ulong_nv,	_atomic_and_ulong_nv)
